Example SAS CODE FOR CONDUCTING SEGMENTED REGRESSION on aggregated data 
 (Steps 1-4) or on individual patient data (Step 5);


*STEP 1.  Create variables for segmented regression;

   *1. create a time variable (here, MONTHS) equal to 1 for 1st time period, and sequentially increasing 
       through both periods;

      data monthdata2;  set monthdata;

         year= admission_year - 2008;   
         months = (year -1)*12 + admission_month;   

   *2. create an intervention variable;

        if admission_year < 2012 then intervention=0;
        else if admission_year >= 2012 then intervention=1;   *intervention began 1/2/2012;

   *3. Create an interaction variable between time variable  and intervention variable such that the 
         intervention variable above estimates the jump in the outcome at start of intervention;

        *for our AMET example, intervention starts at month 37, and below code creates variable
	  "months_minus_36" which is 0 for months < 37 and starts at 1 for month 37, incrementing to the end;


          months_minus_36= max(0, months-36);  *Months after start of intervention 1 (started month 37);


*STEP 2. Aggregate data into appropriate time intervals in each period;|

A. Continuous outcome
  
 *For example, use proc means to calculate and save the average across patients of the outcome 
  [here, HOSP_LOS (hospital length of stay)] and potential confounding variable by month, 
   or relevant time interval;

   months= months from start of study 
   intervention = 0 or 1 (pre-intervention or post-intervention period) 

%let caselevelvars=  HOSP_LOS age bmi comorbidity1-comorbidity100;

   proc sort data= fulldata; by intervention months; 
   proc means data= fulldata  n mean;    by intervention months;
     var &caselevelvars;
     output out=means   mean= &caselevelvars;     run;

  *the above creates new variables of the mean of the outcome and baseline covariates for each month, 
     and keeps the same names as the individual variables (for simplicity);

    Merge data from Steps 1 and 2 into work.aggregated_data

B. Binary Outcome

Use code from Step1A above for covariates, but the binary outcome should be summarized for each time
period in a way which allows either logistic regression in the events/trials format or analysis
on the count of events for the time interval.

Outcome is rapid response encounter (yes/no) – “Rapid_Response”. 
Calculate # events (nevents_RR) and total N (nTotal_RR) for each month.

 proc sort data= fulldata; by intervention months; 
       proc means data= fulldata  n sum;    by intervention months;
           var Rapid_Response;
           output out=response  sum= nevents_RR  N= nTotal_RR;
       run;
    Merge this binary response outcome data with the summary data on covariates in Step 1A. 
    Merge data from Steps 1 and 2 into work.aggregated_data


 *STEP 3. Conduct segmented regression;

   *Choose an appropriate regression model based on the outcome variable and data structure;
    
   *For the AMET data we used logistic regression for the rapid response (RR) variable since it
     is binary.  We used linear regression for a sensitivity analysis on hospital length of stay.


A. Continuous outcome – hospital LOS (sensitivity analysis in AMET study)

*We used the proc GENMOD procedure with identity link (dist=n), as below.
 This is the same as linear regression;

    %let covs=age bmi  Surgical_Encounter RENLFAIL LIVER TUMOR OBESE WGHTLOSS ANEMDEF DRUG DEPRESS;
          proc genmod data= aggregated_data;
	      model outcome= months intervention months_minus_36 &caselevelvars / dist= n; 
            ods output  ParameterEstimates=beta;    
               *saving the beta coefficients and standard errors for calculations;
	    run;
	    *calculate 95% CIs for segmente regression parameters of interest;
          data beta; set beta;
            lowerCL= estimate - 1.96 * stderr;
            upperCL= estimate + 1.96 * stderr;
            pval=probchisq;   *renaming P-value;
	    run;
          title "Segmented Regression results on aggregated data using linear regression";
          proc print data=beta;
          run;



    B. Binary Outcome – rapid response (yes/no)

       We used the LOGISTIC procedure with logit link to conduct logistic regression modeling the 
       #events/#trials for each month as the outcome variable, as below.
   

         %let covs=age bmi  Surgical_Encounter RENLFAIL LIVER TUMOR OBESE WGHTLOSS ANEMDEF DRUG DEPRESS;
         proc logistic data=ed.means_RR_nomiss2 descending;
            model nevents_rr/totalN_rr = months intervention months_minus_36 &covs /rl lackfit link=logit; 
            where months < 80; *The study period ended here;
            output out=preds p=pred   lower=lower_CLM  upper=upper_CLM XBETA=XBETA STDXBETA=se_xBETA  
               stdreschi=stdreschi stdresdev=stdresdev; 
            ods output  ParameterEstimates=beta;  
         run;
         data beta; set beta;
           lowerCL= estimate - 1.96 * stderr;
           upperCL= estimate + 1.96 * stderr;
           pval=probchisq;   *renaming P-value;
	    run;
          title "Segmented Regression results on aggregated data using linear regression";
          proc print data=beta;
          run;




*STEP 4. Compare predicted values under intervention to counterfactual values
         assuming pre-intervention trend;

      *first create separate dataset for each parameter of interest, then merge together below; 
       data b0 b1 b2  b3;  set beta;
         D=1;
         if upcase(parameter)="INTERCEPT" then do;  
            b0= estimate; 
		KEEP B0 D;
		output b0;
        end;
        if upcase(parameter)="MONTHS" then do;
            b1=estimate; 
            b1_lowercl = lowercl;
		b1_uppercl = uppercl;   
            SE_B1= StdErr; 
		KEEP B1 B1_LOWERCL B1_UPPERCL SE_B1 D;
	       ouTput b1;
         end;
         if upcase(parameter)="INTERVENTION" then do;
            b2 =estimate; 
            b2_lowercl = lowercl;
		b2_uppercl = uppercl;
            SE_B2= StdErr; 
		KEEP B2 B2_LOWERCL B2_UPPERCL SE_B2 D;
 	      ouTput b2;
         end;
         if upcase(parameter)="MONTHS_MINUS_36" then do;
            b3 =estimate; 
            b3_lowercl = lowercl;
		b3_uppercl = uppercl;  
            SE_B3= StdErr; 
		KEEP B3 B3_LOWERCL B3_UPPERCL SE_B3 D;
	    ouTput b3;
        end;
     run;
   *Joining estimates for each parameter of interest into a single row for analysis;

    DATA beta_preds; UPDATE b0 b1; BY D;  
    DATA beta_preds; UPDATE beta_preds b2;  BY D;
    DATA beta_preds; UPDATE beta_preds b3;  BY D;  


  *SELECT A STARTTIME INDICATING WHEN THE INTERVENTION BEGAN;                  
     %let starttime=37;  *AMET start month;
  *SELECT A TIME AT WHIFH TO COMPARE INTERVENTIN AND PRE-INTERVENTION TRENDS;  
     %LET PTIME=60;  *COMPARE AFTER 24 MONTHS OF INTERVENTION;
 
    data beta_preds;  length  dependent $30; 
         set beta_preds;


        ptime= &ptime;
[bookmark: _GoBack]        starttime =&starttime;
  
        Y_ptime_with     = b0 + b1 * &ptime  + b2 *1   +  b3 * (&ptime - &starttime);   
          *Equation 3: predicted y at ptime after starting interventi0no at starttime;
        Y_ptime_without =  b0 + b1 * &ptime;       
           *Equation 4: predicted y at ptime without intervention (assumign pre-intercention trend);
        Y_diff_ptime  =  b2 *1  +  b3 * (&ptime - &starttime);   
           *y at ptime --- predicted intervention effect -- equation 2 minus 3 in manuscript;

        Y_diff_var =  1**2 *se_b2 **2  +  ( ((&ptime - &starttime)**2) * se_b3**2) ;   
           *variance of A SUM;  *var (cx + dy) = sum (c2 var(x)  d2 var(y));
        Y_diff_se = sqrt(y_diff_var);

        Y_diff_low95 =  Y_diff_ptime - 1.96 * Y_diff_se;
        Y_diff_up95 =  Y_diff_ptime + 1.96 * Y_diff_se;

        z_stat_diff = Y_diff_ptime/ Y_diff_se;
        P_val = 2*(1- probnorm(abs(z_stat_diff)));

    run;
    TITLE3 "Comparing predicted value of Y at &ptime under  intervention versus w/o intervention";
    proc print dat=beta_preds;
      var  ptime starttime   Y_ptime_with   Y_ptime_without  Y_diff_ptime Y_diff_var Y_diff_se 
         Y_diff_low95   Y_diff_up95  z_stat_diff   p_val;
    run; 

****************************************************************************************;

*STEP 5:  SEGMENTED REGRESSION ON PATIENT-LEVEL DATA;
    
    *For patient level data the actual covariate values are used, if available;   
    *Choose a model that fits the subject level outcome variable (e.g., logistic regression if binary); 
    *If patients represented more than once, adjust for possible within-subject correlation;
    *Otherwise, methods are the same as for the aggregated data;

title "AMET  Segmented regression --RRT yes/no--- patient level";
title2 "ALL OBS per patient -- adjust for correlation";

%let covariates=  age bmi comorbidity1-comorbidity100;
proc genmod data=AMET  DESCENDING;
   class MRN; 
   model Rapid_Response  = months intervention months_minus_36   &covariates / dist=b; 
       *use logistic regression since outcome is binary;
   repeated subject=MRN / type=cs;  *assume exchangeable correlation;  
       *repeated statement uses GEE model adjusting for within-subject correlation;
 
   ods output   GEEEmpPEst =beta;    *same betas and robust variance-covariance estimates from GEE;
run;
data beta; set beta;
   lowerCL= estimate - 1.96 * stderr;
   upperCL= estimate + 1.96 * stderr;

   OR=exp(estimate);   *odds ratio for each parameter of interest;
   OR_low= exp(estimate - 1.96*stderr);
   OR_up = exp(estimate + 1.96*stderr);
   
run;


SAS Code for Autocorrelation analyses.

In SAS an autocorrelation-adjusted segmented regression analysis could be done using the AUTOREG procedure in the ETS module, as below.
SAS code for adjusting for autocorrelation in segmented regression using continuous outcome.

            proc autoreg data=a;
   Model   y = time   intervention   time_after_intervention  / nlag=1 method=ml;
            run;
            The NLAG=1  option is used to estimate the first-order auto-correlation estimation and adjustment.

